Beltrami framework

Marc Golub, 2018

Introduction

The purpose of this notebook is to help compute the symbolic form of some cumbersome expressions used in a Beltrami framework algorithm that fits a bi-tensor model to single-shell DTI datasets, allowing (hopefully) for the extraction of free water maps and corrected diffusion measures. The algorithm is based on the work of Pasternak et al (2009) [1].

Iwasawa parameters

The 6 independent diffusion parameters D are related to the Iwasawa parameters X by:

In[609]:=
```
eq1 = Dxx == X1;
eq2 = Dyy == X2 + X1 * X4^2;
eq3 = Dzz == X3 + X1 * X5^2 + X2 * X6^2;
eq4 = Dxy == X1 * X4;
eq5 = Dxz == X1 * X5;
eq6 = Dyz == X1 * X4 * X5 + X2 * X6;
sol = Solve[{eq1, eq2, eq3, eq4, eq5, eq6}, {X1, X2, X3, X4, X5, X6}];
For[i = 1, i ≤ 6, i++,
 res = sol[[1, i]];
 Print[FullSimplify[res]]
]
```

$X_1 \to D_{xx}$

$X_2 \to -\frac{D_{xy}^2}{D_{xx}} + D_{yy}$

$X_3 \to \frac{D_{xz}^2 D_{yy} - 2 D_{xy} D_{xz} D_{yz} + D_{xx} D_{yz}^2}{D_{xy}^2 - D_{xx} D_{yy}} + D_{zz}$

$X_4 \to \frac{D_{xy}}{D_{xx}}$

$X_5 \to \frac{D_{xz}}{D_{xx}}$

$X_6 \to \frac{D_{xy} D_{xz} - D_{xx} D_{yz}}{D_{xy}^2 - D_{xx} D_{yy}}$

In[617]:=
```
Clear[i]
```

The spatial-feature manifold metric h

The h metric is a 9 by 9 matrix, constructed with the Iwasawa parameters, where β determines the ratio between the feature space and image domain distances.

In[618]:=
```
h44 = β * 1 / X1^2;
h55 = β * 1 / X2^2;
h66 = β * 1 / X3^2;
h77 = β * (2 * X1 * (X3 + X2 * X6^2) / (X2 * X3));
h78 = β * (-2 * X1 * X6) / X3;
h88 = β * 2 * X1 / X3;
h99 = β * 2 * X2 / X3;
```

In[625]:=
```
h = {
   {1, 0, 0, 0, 0, 0, 0, 0, 0},
   {0, 1, 0, 0, 0, 0, 0, 0, 0},
   {0, 0, 1, 0, 0, 0, 0, 0, 0},
   {0, 0, 0, h44, 0, 0, 0, 0, 0},
   {0, 0, 0, 0, h55, 0, 0, 0, 0},
   {0, 0, 0, 0, 0, h66, 0, 0, 0},
   {0, 0, 0, 0, 0, 0, h77, h78, 0},
   {0, 0, 0, 0, 0, 0, h78, h88, 0},
   {0, 0, 0, 0, 0, 0, 0, 0, h99}
  };
MatrixForm[Together[h]]
```

Out[626]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{\beta}{X_1^2} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{\beta}{X_2^2} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{\beta}{X_3^2} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \frac{2\beta X_1 (X_3 + X_2 X_6^2)}{X_2 X_3} & -\frac{2\beta X_1 X_6}{X_3} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -\frac{2\beta X_1 X_6}{X_3} & \frac{2\beta X_1}{X_3} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{2\beta X_2}{X_3} \end{pmatrix}$$

In[627]:=
```
Clear[h, h44, h55, h66, h77, h78, h88, h99]
```

The image manifold metric γ

The γ metric is related to the h metric and spatial derivatives of X by:

$$\gamma_{\mu\nu} = dX_{\mu i}\, dX_{\nu j}\, h_{ij}$$

Where dX is the 3 by 9 derivatives matrix.

In[628]:=
```
dX = {
   {1, 0, 0, dX1x, dX2x, dX3x, dX4x, dX5x, dX6x},
   {0, 1, 0, dX1y, dX2y, dX3y, dX4y, dX5y, dX6y},
   {0, 0, 1, dX1z, dX2z, dX3z, dX4z, dX5z, dX6z}
  };
```

In[629]:=
```
MatrixForm[dX]
```

Out[629]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & dX1x & dX2x & dX3x & dX4x & dX5x & dX6x \\ 0 & 1 & 0 & dX1y & dX2y & dX3y & dX4y & dX5y & dX6y \\ 0 & 0 & 1 & dX1z & dX2z & dX3z & dX4z & dX5z & dX6z \end{pmatrix}$$

Where dX1x is the derivative of X1 with respect to x, dX2y is the derivative of X2 with respect to y and so on...

In[630]:=
```
h = {
    {1, 0, 0, 0, 0, 0, 0, 0, 0},
    {0, 1, 0, 0, 0, 0, 0, 0, 0},
    {0, 0, 1, 0, 0, 0, 0, 0, 0},
    {0, 0, 0, β * h44, 0, 0, 0, 0, 0},
    {0, 0, 0, 0, β * h55, 0, 0, 0, 0},
    {0, 0, 0, 0, 0, β * h66, 0, 0, 0},
    {0, 0, 0, 0, 0, 0, β * h77, β * h78, 0},
    {0, 0, 0, 0, 0, 0, β * h78, β * h88, 0},
    {0, 0, 0, 0, 0, 0, 0, 0, β * h99}
  };
g = FullSimplify[Table[Sum[dX[[μ, i]] * dX[[ν, j]] * h[[i, j]],
     {i, 1, 9}, {j, 1, 9}], {μ, 1, 3}, {ν, 1, 3}]];
StringForm["γ11 = ``", g[[1, 1]]]
StringForm["γ22 = ``", g[[2, 2]]]
StringForm["γ33 = ``", g[[3, 3]]]
StringForm["γ12 = ``", g[[1, 2]]]
StringForm["γ13 = ``", g[[1, 3]]]
StringForm["γ23 = ``", g[[2, 3]]]
```

Out[632]= γ_{11} =

$1 + \left(dX1x^2\, h44 + dX2x^2\, h55 + dX3x^2\, h66 + dX4x^2\, h77 + 2\, dX4x\, dX5x\, h78 + dX5x^2\, h88 + dX6x^2\, h99\right) \beta$

Out[633]= γ_{22} =

$1 + \left(dX1y^2\, h44 + dX2y^2\, h55 + dX3y^2\, h66 + dX4y^2\, h77 + 2\, dX4y\, dX5y\, h78 + dX5y^2\, h88 + dX6y^2\, h99\right) \beta$

Out[634]= γ_{33} =

$1 + \left(dX1z^2\, h44 + dX2z^2\, h55 + dX3z^2\, h66 + dX4z^2\, h77 + 2\, dX4z\, dX5z\, h78 + dX5z^2\, h88 + dX6z^2\, h99\right) \beta$

Out[635]= γ_{12} =

$\left(dX1x\, dX1y\, h44 + dX2x\, dX2y\, h55 + dX3x\, dX3y\, h66 + dX4x\, dX4y\, h77 + dX4y\, dX5x\, h78 + dX4x\, dX5y\, h78 + dX5x\, dX5y\, h88 + dX6x\, dX6y\, h99\right) \beta$

Out[636]= γ_{13} =

$\left(dX1x\, dX1z\, h44 + dX2x\, dX2z\, h55 + dX3x\, dX3z\, h66 + dX4x\, dX4z\, h77 + dX4z\, dX5x\, h78 + dX4x\, dX5z\, h78 + dX5x\, dX5z\, h88 + dX6x\, dX6z\, h99\right) \beta$

Out[637]= γ_{23} =

$\left(dX1y\, dX1z\, h44 + dX2y\, dX2z\, h55 + dX3y\, dX3z\, h66 + dX4y\, dX4z\, h77 + dX4z\, dX5y\, h78 + dX4y\, dX5z\, h78 + dX5y\, dX5z\, h88 + dX6y\, dX6z\, h99\right) \beta$

The Christoffel symbols

The Christoffel symbols are used to compute the Levi-Civita term and are given by:

$$[\Gamma_{jk}]^{i} = \frac{1}{2} h^{il} \left(\partial_j h_{lk} + \partial_k h_{jl} - \partial_l h_{jk}\right)$$

Where h^{il} is the inverse of h and ∂_j denotes the symbolic partial derivative with respect to X_j. Note that the expression above might be a correction of the expression given in the original article. From now on let h be a sliced version of the original metric, with only the important indices (from 4 to 9), with dimension:

In[638]:=
```
n = 6;
```

In[639]:=
```
Clear[coord, h, i, j, k, l]
```

In[640]:=
```
coord = {X₁, X₂, X₃, X₄, X₅, X₆};
```

In[641]:=
```
h = {
   {β * 1 / X₁^2, 0, 0, 0, 0, 0},
   {0, β * 1 / X₂^2, 0, 0, 0, 0},
   {0, 0, β * 1 / X₃^2, 0, 0, 0},
   {0, 0, 0, β * 2 * X₁ * (X₃ + X₂ * X₆^2) / (X₂ * X₃), -β * 2 * X₁ * X₆ / X₃, 0},
   {0, 0, 0, -β * 2 * X₁ * X₆ / X₃, β * 2 * X₁ / X₃, 0},
   {0, 0, 0, 0, 0, β * 2 * X₂ / X₃}};
MatrixForm[h]
```

Out[642]//MatrixForm=

$$\begin{pmatrix} \frac{\beta}{X_1^2} & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{\beta}{X_2^2} & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{\beta}{X_3^2} & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{2\,\beta\,X_1\,(X_3+X_2\,X_6^2)}{X_2\,X_3} & -\frac{2\,\beta\,X_1\,X_6}{X_3} & 0 \\ 0 & 0 & 0 & -\frac{2\,\beta\,X_1\,X_6}{X_3} & \frac{2\,\beta\,X_1}{X_3} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{2\,\beta\,X_2}{X_3} \end{pmatrix}$$

The inverse is given by:

In[643]:=
```
hinv = Simplify[Inverse[h]];
MatrixForm[hinv]
```

Out[644]//MatrixForm=

$$\begin{pmatrix} \frac{X_1^2}{\beta} & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{X_2^2}{\beta} & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{X_3^2}{\beta} & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{X_2}{2\,\beta\,X_1} & \frac{X_2\,X_6}{2\,\beta\,X_1} & 0 \\ 0 & 0 & 0 & \frac{X_2\,X_6}{2\,\beta\,X_1} & \frac{X_3+X_2\,X_6^2}{2\,\beta\,X_1} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{X_3}{2\,\beta\,X_2} \end{pmatrix}$$

The Christoffel symbols are computed by using the definition. The following code was adapted from a notebook used to teach a physics course, by James B. Hartle [2].

In[645]:=
```
n = 6
```
Out[645]=
```
6
```
In[646]:=
```
affine := affine = Simplify[Table[(1/2) * Sum[(hinv[[i, l]]) *
        (D[h[[l, k]], coord[[j]] ] +
          D[h[[j, l]], coord[[k]] ] - D[h[[j, k]], coord[[l]] ]), {l, 1, n}],
     {i, 1, n}, {j, 1, n}, {k, 1, n}] ]
```
In[647]:=
```
listaffine := Table[If[UnsameQ[affine[[i, j, k]], 0],
    {ToString[Γ[i, j, k]], affine[[i, j, k]]}] , {i, 1, n}, {j, 1, n}, {k, 1, j}]
```

In[648]:=
```
TableForm[Partition[DeleteCases[Flatten[listaffine], Null], 2],
  TableSpacing → {2, 2}]
```

Out[648]//TableForm=

Γ[1, 1, 1]	$-\frac{1}{X_1}$
Γ[1, 4, 4]	$-\frac{X_1^2 (X_3+X_2 X_6^2)}{X_2 X_3}$
Γ[1, 5, 4]	$\frac{X_1^2 X_6}{X_3}$
Γ[1, 5, 5]	$-\frac{X_1^2}{X_3}$
Γ[2, 2, 2]	$-\frac{1}{X_2}$
Γ[2, 4, 4]	X_1
Γ[2, 6, 6]	$-\frac{X_2^2}{X_3}$
Γ[3, 3, 3]	$-\frac{1}{X_3}$
Γ[3, 4, 4]	$X_1 X_6^2$
Γ[3, 5, 4]	$-X_1 X_6$
Γ[3, 5, 5]	X_1
Γ[3, 6, 6]	X_2
Γ[4, 4, 1]	$\frac{1}{2 X_1}$
Γ[4, 4, 2]	$-\frac{1}{2 X_2}$
Γ[4, 6, 4]	$\frac{X_2 X_6}{2 X_3}$
Γ[4, 6, 5]	$-\frac{X_2}{2 X_3}$
Γ[5, 4, 2]	$-\frac{X_6}{2 X_2}$
Γ[5, 4, 3]	$\frac{X_6}{2 X_3}$
Γ[5, 5, 1]	$\frac{1}{2 X_1}$
Γ[5, 5, 3]	$-\frac{1}{2 X_3}$
Γ[5, 6, 4]	$\frac{1}{2}\left(-1+\frac{X_2 X_6^2}{X_3}\right)$
Γ[5, 6, 5]	$-\frac{X_2 X_6}{2 X_3}$
Γ[6, 4, 4]	$-\frac{X_1 X_6}{X_2}$
Γ[6, 5, 4]	$\frac{X_1}{2 X_2}$
Γ[6, 6, 2]	$\frac{1}{2 X_2}$
Γ[6, 6, 3]	$-\frac{1}{2 X_3}$

There are 26 unique non zero Christoffel symbols.

References

- Pasternak, O., Sochen, N., Gur, Y., Intrator, N., & Assaf, Y. (2009). Free water elimination and mapping from diffusion mri. Magnetic resonance in medicine, 62 (3), 717–730. [1]
- Christoffel Symbols and Geodesic Equation. Available online at http://web.physics.ucsb.edu/~gravitybook/mathematica.html. [2]